10.3 Complex Fourier Transform

A. Purpose

This subroutine computes Fourier transforms for complex data in up to 6 dimensions using the fast Fourier transform. The relation between values z and Fourier coefficients ζ is defined by

$$z(j_1, j_2, \ldots, j_{ND}) = \sum_{k_1=0}^{N_1-1} \cdots \sum_{k_{ND}=0}^{N_{ND}-1} \zeta(k_1, k_2, ..., k_{ND}) W_1^{j_1 k_1} \cdots W_{ND}^{j_{ND} k_{ND}}, \text{ and}$$

$$\zeta(k_1, k_2, \ldots, k_{ND}) = \frac{1}{N_1} \cdots \frac{1}{N_{ND}} \sum_{j_1=0}^{N_1-1} \cdots \sum_{j_{ND}=0}^{N_{ND}-1} z(j_1, \ldots, j_{ND}) W^{-j_1 k_1} \cdots W^{-j_{ND} k_{ND}}$$

where $N_\ell = 2^{\mathrm{M}(\ell)}$, $W_\ell = e^{2\pi i/N_\ell}$, $0 \le j_\ell,\ k_\ell \le N_\ell - 1$, and z and ζ are complex.

B. Usage

B.1 Program Prototype, Single Precision

COMPLEX **A**(N_1, N_2, ..., $\ge N_{ND}$) $[N_k = 2^{\mathrm{M}(k)}]$

REAL **S**($\ge \max(\nu_1, \nu_2, ..., \nu_{ND}) - 1$) $[\nu_k = 2^{\mathrm{M}(k)-2}]$

INTEGER **M**($\ge ND$), **ND, MS**

CHARACTER **MODE***($\ge ND$)

On the initial call set MS to 0 to indicate the array S() does not yet contain a sine table. Assign values to A(), MODE, M, and ND.

CALL SCFT(A, MODE, M, ND, MS, S)

A() will contain computed results. S() will contain the sine table used in computing the Fourier transform. MS may have been changed.

B.2 Argument Definitions

A() [inout] If the argument MODE selects analysis in all dimensions, A() contains values z on entry, and Fourier coefficients ζ on exit. If MODE selects analysis in all dimensions, A() contains Fourier coefficients ζ on entry, and values z on exit. When A() contains z, A($j_1 + 1$, $j_2 + 1$, ..., $j_{ND} + 1$) = z(j_1, j_2, ..., j_{ND}), and when A() contains ζ, A($k_1 + 1$, $k_2 + 1$, ..., $k_{ND} + 1$) = ζ(k_1, k_2, ..., k_{ND}), $0 \le j_i, k_i \le 2^{\mathrm{M}(i)} - 1$, $i = 0, 1, ..., \mathrm{ND}$.

MODE [in] The character MODE(k:k) selects Analysis or Synthesis in the k^{th} dimension. 'A' or 'a' selects Analysis, transforming z's to ζ's. 'S' or 's' selects Synthesis, transforming ζ's to $z's$.

M() [in] Defines $N_k = 2^{\mathrm{M}(k)}$, the number of complex data points in the k^{th} dimension. Require $0 \le \mathrm{M}(k) \le 30$ for all k. No action is taken in dimensions for which $\mathrm{M}(k) = 0$.

ND [in] Number of dimensions. Require $1 \le \mathrm{ND} \le 6$.

MS [inout] Gives the state of the sine table in $S()$. Let MS_{in} and MS_{out} denote the values of MS on entry and return respectively. If the sine table has not previously been computed, set $\mathrm{MS}_{in} = 0$ or -1 before the call. Otherwise the value of MS_{out} from the previous call using the same S() array can be used as MS_{in} for the current call.

Certain error conditions described in Section E cause the subroutine to set $\mathrm{MS}_{out} = -2$ and return. Otherwise, with $\max_i\{\mathrm{M}(i)\} > 0$, the subroutine sets $\mathrm{MS}_{out} = \max(\mathrm{M}(1), \mathrm{M}(2), ..., \mathrm{M(ND)}, \mathrm{MS}_{in})$.

If $\mathrm{MS}_{out} > \max(2, \mathrm{MS}_{in})$, the subroutine sets $\mathrm{NT} = 2^{\mathrm{MS}_{out}-2}$ and fills S() with NT − 1 sine values.

If $\mathrm{MS}_{in} = -1$, the subroutine returns after the above actions, not transforming the data in A(). This is intended to allow the use of the sine table for data alteration before a subsequent Fourier transform, as discussed in Section G of Chapter 16.0.

S() [inout] When the sine table has been computed, $\mathrm{S}(j) = \sin \pi j/(2 \times NT)$, $j = 1, 2, ..., \mathrm{NT} - 1$, see MS above.

B.3 Modifications for Double Precision

Change SCFT to DCFT and the REAL type statement to DOUBLE PRECISION. If it is available, one can change the COMPLEX type statement to DOUBLE PRECISION COMPLEX. For portability, (or out of necessity) one can change the COMPLEX statement to DOUBLE PRECISION and change the first dimension to be twice as big. The data should then be stored in A with the imaginary parts of the complex numbers following immediately after the real parts. This representation is compatible with the representation used in Fortran 90, and with most compilers that extend Fortran 77 to provide a double precision complex type.

C. Examples and Remarks

Estimate the spectral composition of

$$f(t) = [\sin 2\pi(t + 0.1) + 4 \cos 2\pi(\sqrt{2}\, t + 0.3)] + 0i$$

©1997 Calif. Inst. of Technology, 2015 Math à la Carte, Inc.

where we make the same assumptions and use the same Δt and N as in the example for SRFT1. Differences between the results given here and those obtained for SRFT1 are due to the use of sigma factors in SRFT1. (It is more efficient to use SRFT1 when f is a real function. A real function was used here to show the effect of the sigma factors.) Note that the peaks are slightly sharper here than they are for SRFT1, but that as one leaves the peaks the coefficients do not tend to zero nearly as rapidly as when the sigma factors are used. The program to do these calculations and the results are given at the end of this chapter.

D. Functional Description

The multi-dimensional complex transform involves calling SFFT to compute one-dimensional complex transforms with respect to each dimension. For ND = 1, the formulas are given by Eqs. (9) and (10) in Chapter 16.0. For $ND \geq 1$, the formula for z given ζ is given in Purpose above. More details can be found in [1].

References

1. Fred T. Krogh, **CFT — Multi-dimensional Complex Fourier Transform**. TU Doc. CP-2312, NPO 11651, Jet Propulsion Laboratory, Pasadena, CA (1970).

E. Error Procedures and Restrictions

Require $1 \leq \text{ND} \leq 6$ and $0 \leq \text{M}(k) \leq 30$ for all k. MODE must have one of its allowed values. If any of these conditions are violated, the subroutine will issue an error message using the error processing procedures of Chapter 19.2 with severity level 2 to cause execution to stop. A return is made with MS = −2 instead of stopping if the statement "CALL ERMSET(−1)" is executed before calling this subroutine.

If the sine table does not appear to have valid data, an error message is printed, and the sine table and then the transform are computed.

F. Supporting Information

The source language is ANSI Fortran 77.

Entry	Required Files
DCFT	DCFT, DFFT, ERFIN, ERMSG, IERM1, IERV1
SCFT	ERFIN, ERMSG, IERM1, IERV1, SCFT, SFFT

Subroutine designed and written by: Fred T. Krogh, JPL, October 1969, revised January 1988.

DRSCFT

```
      program DRSCFT
c>> 1996-06-05 DRSCFT Krogh Fixes for conversion to C.
c>> 1994-10-19 DRSCFT Krogh  Changes to use M77CON
c>> 1989-05-07 DRSCFT FTK, CLL
c     Driver to demonstrate SCFT -- Complex Fast Fourier Transform
c     ------------------------------------------------------------------
c--S replaces "?": DR?CFT, ?CFT
c     ------------------------------------------------------------------
      real               F(256)
      real              S(31), T, DELTAT, TTIME, PI, TWOPI, RTWO
      real               ZERO, FOUR, P1, P3
      integer     I, K, KF, M, MS, N, ND, MA(1)
      parameter (PI = 3.14159265358979323 84E0)
      parameter (TWOPI = 2.E0 * PI)
      parameter (RTWO = 1.41421356237309504 88E0)
      parameter (M = 7)
c       parameter (N = 2**M)
      parameter (N = 128)
      parameter (ND = 1)
      parameter (ZERO = 0.E0)
      parameter (FOUR = 4.E0)
      parameter (P1 = .1E0)
      parameter (P3 = .3E0)
      data  TTIME /25.E0/
      data MA / M /
c     ------------------------------------------------------------------
      print*,'Program DRSCFT..  Demonstrates SCFT'
      DELTAT = TTIME / N
      T = ZERO
      do 100 K = 1, 2*N, 2
         F(K)=SIN(TWOPI*(T+P1))+FOUR*COS(TWOPI*(RTWO*T+P3))
         F(K+1) = ZERO
         T = T + DELTAT
  100 continue
      MS = 0
      call SCFT (F, 'A', MA, ND, MS, S)
      print'(1X/A/1X)',' TRANSFORM FOR K=21 TO 41 (NO SIGMA FACTORS)'
      do 120 K = 21, 41, 3
         KF = 2*K-1
         print'(A,I3,A,I3,3(F11.5,F9.5))',
     *        ' K=',K,' TO',K+2,(F(I), I=KF, KF+5)
  120 continue
      stop
      end
```

ODSCFT

```
 Program DRSCFT..  Demonstrates SCFT

 TRANSFORM FOR K=21 TO 41 (NO SIGMA FACTORS)

 K= 21 TO 23    -0.03419  0.00454   -0.03652  0.00497   -0.03924  0.00545
 K= 24 TO 26    -0.04244  0.00600   -0.04624  0.00663    0.24306 -0.39712
 K= 27 TO 29    -0.05645  0.00827   -0.06347  0.00936   -0.07246  0.01073
 K= 30 TO 32    -0.08435  0.01250   -0.10076  0.01492   -0.12482  0.01841
 K= 33 TO 35    -0.16334  0.02395   -0.23478  0.03414   -0.41197  0.05926
 K= 36 TO 38    -1.58776  0.22542    0.88496 -0.12372    0.35101 -0.04820
 K= 39 TO 41     0.22103 -0.02975    0.16252 -0.02138    0.12932 -0.01659
```